


# How To Make Simple Geometries in Gmsh

(By: Nick Battista 9/24/14)

This tutorial will guide you how to make a **tube** in *gmsh*. We will introduce how to use basic **parameters**, **points**, **circles**, **line segments**, **ruled surfaces**, and **plane surfaces**.


1. First we open gmsh. This is the first obstacle -> Getting the file to save where you want it to. Before doing anything go to FILE -> SAVE AS and then choose the appropriate folder and save it as geometry.geo

Now in the gmsh window, click the *Modules* and then *Geometry* tab. Now look down through the window and click "Reload."



2. Next we will add some *parameters* that we can use in constructing the geometry, i.e., we will want our tube to have a specific length,  $L$ , so we can define  $L$ .

To do this click the *Elementary Entities* tab. You will now see a window with the following:



This list basically has a lot of the geometric entities we can use at our disposal to construct the mesh.

We can define parameters

Note that we can add:



1. Individual Points
2. Line Segments (between pts.)
3. Circle/Circular Arc
4. Ellipse/Elliptic Arc
5. Splines

Then to define our mesh, we can define a surface as in a flat surface (*plane surface*) or curved surface (*ruled surface*)

3. Next we want to add some parameters. We will add the following parameters in constructing our straight tube geometry.


- a.  $L$  (length of tube)
- b.  $r$  (radius of tube)

To add some parameters, click *Parameters*. We want to define  $L$  to be equal to 2.0 and  $r$  to be 0.25. Enter the appropriate values, like in the pictures below, and click add.



4. We will now go open a text-like file, which we can effectively write the whole mesh geometry in and continually *Reload* the file to see the updates to the geometric mesh on the screen. Note: Once you get the hang of gmsh, this will be your best friend.

To open the text file, click *Edit file*. You will see




NOTE: For **Windows** users : Go to *Tools -> Options -> General -> Advanced*. Change the text editor command from "notepad.exe %s" to "C:/Windows/notepad.exe %s"

In this file, we can add points, line segments, surfaces, etc, by manually typing them in.


For example say we want to make a new parameter, called  $ds$ , that is equal to  $L/100$  and will be our "element spacing"

We can do this by writing this line in that file:


```
ds= DefineNumber[ L/100.0, Name "Parameters/r" ];
```



- Now we will begin adding a few points! These points will essentially be at the poles (as in N,S,E or W) of our circle (looking at the circular head of the tube straight-on).



To add the points, click *Point* and enter the information for Pt. 1. (Note: add *ds* to 'prescribed mesh element size at point') and then click *Add*.



Now go back and open the text-file, by clicking *Edit file*. You should now see:

```
untitled.geo
L = DefineNumber[ 2.0, Name "Parameters/L" ];
r = DefineNumber[ 0.25, Name "Parameters/r" ];
ds = DefineNumber[ L/100, Name "Parameters/ds" ];

Point(1) = {0, r, L/2, ds};
```

Next we want to add the four other points (**in order**). We could go back to the gui and add each point by adding the individual information, or you can just do it all in this file. Type the following:


```

untitled.geo
L = DefineNumber[ 2.0, Name "Parameters/L" ];
r = DefineNumber[ 0.25, Name "Parameters/r" ];
ds = DefineNumber[ L/100, Name "Parameters/ds" ];

Point(1) = {0, r, L/2, ds};
Point(2) = {-r,0, L/2, ds};
Point(3) = {0, -r, L/2, ds};
Point(4) = {r, 0, L/2, ds};
Point(5) = {0, 0, L/2, ds};


```

Finally click on *Reload* to make the changes. In this mesh-geometry box, you should now see those 5 points, i.e.,



6. Now let's connect the "pole" points using circular arcs. To do this click *Circle Arc*. Now to define an arc, you now need to click on 3-points in sequence.
  - a. The first point is the point where the arc starts.
  - b. The second point is the central point for the circular arc
  - c. The third point is where the arc ends

To add an arc between points 1 and 2, click on Point 1, then Point 5, then Point 2. You should now see an arc. (Note when you click on the points, they should highlight in red.) Now go and open the text-file again!



```

untitled.geo
L = DefineNumber[ 2.0, Name "Parameters/L" ];
r = DefineNumber[ 0.25, Name "Parameters/r" ];
ds = DefineNumber[ L/100, Name "Parameters/ds" ];

Point(1) = {0, r, L/2, ds};
Point(2) = {-r,0, L/2, ds};
Point(3) = {0, -r, L/2, ds};
Point(4) = {r, 0, L/2, ds};
Point(5) = {0, 0, L/2, ds};

Circle(1) = {1, 5, 2};
|

```

- Now to add the rest of the circular arcs, let just do it in the text-file (or you can keep clicking the points if you'd like). Remember the order of the points matter.


Also, I am doing everything going counter-clockwise. It's good just to keep a constant convention in this.

```
untitled.geo
L = DefineNumber[ 2.0, Name "Parameters/L" ];
r = DefineNumber[ 0.25, Name "Parameters/r" ];
ds = DefineNumber[ L/100, Name "Parameters/ds" ];

Point(1) = {0, r, L/2, ds};
Point(2) = {-r,0, L/2, ds};
Point(3) = {0, -r, L/2, ds};
Point(4) = {r, 0, L/2, ds};
Point(5) = {0, 0, L/2, ds};

Circle(1) = {1, 5, 2};
Circle(2) = {2, 5, 3};
Circle(3) = {3, 5, 4};
Circle(4) = {4, 5, 1};
```

- After you've saved and *Reload*-ed the file, you should now see a complete circle.



Now we want to add another circle at  $z = -L/2$  to construct our tube. We will do this completely in the text-file (as seen below)! In the geometry window, you can click and rotate to see the two circles.

```

untitled.geo
L = DefineNumber[ 2.0, Name "Parameters/L" ];
r = DefineNumber[ 0.25, Name "Parameters/r" ];
ds = DefineNumber[ L/100, Name "Parameters/ds" ];


Point(1) = {0, r, L/2, ds};
Point(2) = {-r, 0, L/2, ds};
Point(3) = {0, -r, L/2, ds};
Point(4) = {r, 0, L/2, ds};
Point(5) = {0, 0, L/2, ds};

Circle(1) = {1, 5, 2};
Circle(2) = {2, 5, 3};
Circle(3) = {3, 5, 4};
Circle(4) = {4, 5, 1};

Point(6) = {0, r, -L/2, ds};
Point(7) = {-r, 0, -L/2, ds};
Point(8) = {0, -r, -L/2, ds};
Point(9) = {r, 0, -L/2, ds};
Point(10) = {0, 0, -L/2, ds};

Circle(5) = {6, 10, 7};
Circle(6) = {7, 10, 8};
Circle(7) = {8, 10, 9};
Circle(8) = {9, 10, 6};

```



9. Next we will add a few line segment between the 'poles' on the circles, i.e., attach a line segment between Point 1 and 6, 2 and 7, 3 and 8, and 4 and 9.

To do this click on *Straight Line* and then click on the points where you want the line segment to attach to. First click on Point 1 and then Point 6. Then click on Point 2 and click on Point 7. You should now see two lines. NOTE: if you want to stop adding lines between points now in your window (i.e., want free clicking control without consequences) click 'q' to end the *Straight Line* mode. This is similar for circular arcs, elliptic arcs, etc.)

```

untitled.geo
L = DefineNumber[ 2.0, Name "Parameters/L" ];
r = DefineNumber[ 0.25, Name "Parameters/r" ];
ds = DefineNumber[ L/100, Name "Parameters/ds" ];

Point(1) = {0, r, L/2, ds};
Point(2) = {-r, 0, L/2, ds};
Point(3) = {0, -r, L/2, ds};
Point(4) = {r, 0, L/2, ds};
Point(5) = {0, 0, L/2, ds};


Circle(1) = {1, 5, 2};
Circle(2) = {2, 5, 3};
Circle(3) = {3, 5, 4};
Circle(4) = {4, 5, 1};

Point(6) = {0, r, -L/2, ds};
Point(7) = {-r, 0, -L/2, ds};
Point(8) = {0, -r, -L/2, ds};
Point(9) = {r, 0, -L/2, ds};
Point(10) = {0, 0, -L/2, ds};

Circle(5) = {6, 10, 7};
Circle(6) = {7, 10, 8};
Circle(7) = {8, 10, 9};
Circle(8) = {9, 10, 6};

Line(9) = {1, 6};
Line(10) = {2, 7};

```




To add the other two lines, let's just do it in the text-file! You only need to add the lines

```
Line(11) = {3, 8};  
Line(12) = {4, 9};
```


Note that even though it says Line(12), we have only defined 4 lines so far. This is because for every geometric construction besides points (circle, elliptic arcs, lines, etc) the counter doesn't discriminate between them. For us we have made 8 circular arcs so far, and then put in 4 line segments, so that is why we are at 12 now.

After saving, you should now see.




10. We are finally at the point to start adding surfaces! (Or actually constructing the mesh!) First let's define the curved surface.

Click on *Ruled Surface*. Then in sequence click on Circle(1) then Line(9) then Circle(5) and then Line(10). After you've clicked everything you should see the surface you're interested in enclosed in red. To finish defining the surface, click 'e'.



After pressing 'e'.

Note the dotted lines  
signifying a surface.



Now you can similarly keep adding those *Ruled Surfaces* in a similar manner,  
or we can do everything in the text-file.

```

untitled.geo
L = DefineNumber[ 2.0, Name "Parameters/L" ];
r = DefineNumber[ 0.25, Name "Parameters/r" ];
ds = DefineNumber[ L/100, Name "Parameters/ds" ];

Point(1) = {0, r, L/2, ds};
Point(2) = {-r, 0, L/2, ds};
Point(3) = {0, -r, L/2, ds};
Point(4) = {r, 0, L/2, ds};
Point(5) = {0, 0, L/2, ds};

Circle(1) = {1, 5, 2};
Circle(2) = {2, 5, 3};
Circle(3) = {3, 5, 4};
Circle(4) = {4, 5, 1};


Point(6) = {0, r, -L/2, ds};
Point(7) = {-r, 0, -L/2, ds};
Point(8) = {0, -r, -L/2, ds};
Point(9) = {r, 0, -L/2, ds};
Point(10) = {0, 0, -L/2, ds};

Circle(5) = {6, 10, 7};
Circle(6) = {7, 10, 8};
Circle(7) = {8, 10, 9};
Circle(8) = {9, 10, 6};

Line(9) = {1, 6};
Line(10) = {2, 7};
Line(11) = {3, 8};
Line(12) = {4, 9};


Line Loop(13) = {1, 10, -5, -9};
Ruled Surface(14) = {13};
Line Loop(15) = {2, 11, -6, -10};
Ruled Surface(16) = {15};
Line Loop(17) = {3, 12, -7, -11};
Ruled Surface(18) = {17};
Line Loop(19) = {4, 9, -8, -12};
Ruled Surface(20) = {19};

```






11. Finally to finish constructing the mesh, click on the *Mesh* tab and then *2D*.



12. Now if you'd like to make a closed tube. We can essentially do the same thing for the circular faces, except use *Plane Surface*.

Click *Plane Surface* and then click Circle(1) then Circle(2) then Circle(3) and then Circle(4). The enclosed plane should now be in red. Click 'e' to finish defining it as a surface.



```

Circle(4) = {4, 5, 1};

Point(6) = {0, r, -L/2, ds};
Point(7) = {-r, 0, -L/2, ds};
Point(8) = {0, -r, -L/2, ds};
Point(9) = {r, 0, -L/2, ds};
Point(10) = {0, 0, -L/2, ds};

Circle(5) = {6, 10, 7};
Circle(6) = {7, 10, 8};
Circle(7) = {8, 10, 9};
Circle(8) = {9, 10, 6};


Line(9) = {1, 6};
Line(10) = {2, 7};
Line(11) = {3, 8};
Line(12) = {4, 9};

Line Loop(13) = {1, 10, -5, -9};
Ruled Surface(14) = {13};
Line Loop(15) = {2, 11, -6, -10};
Ruled Surface(16) = {15};
Line Loop(17) = {3, 12, -7, -11};
Ruled Surface(18) = {17};
Line Loop(19) = {4, 9, -8, -12};
Ruled Surface(20) = {19};



Line Loop(21) = {1, 2, 3, 4};
Plane Surface(22) = {21};

```

Now simply, go to the *Mesh* tab and click *2D* again. You should now see the following



## DIFFERENT VIEWS OF HALF-CLOSED TUBE



**Open End**

**Closed End**

